EVALUATION OF GAMMA SCORE

By J. ROY

Indian Statistical Institute

SUMMARY : Mean values of specified fractile intervals of the standard normal distribution were called normal scores and tabulated by Fisher and Yates. Mean values of fractile intervals of the single parameter gamma distribution will be called *gamma-scores*. Gamma scores are useful in examining the goodness of fit of a gamma distribution by fractile graphical methods developed by Mahalanobis.

Numerical methods and computer programs for evaluation of gamma-scores are presented in this paper.

1. Introduction

Let $F(x)$ be the distribution function of a random variable whose expectation exists. We assume that for any given p, $0 < p < 1$, the equation $F(x) = p$ has a unique solution which we shall denote by $\xi(p)$. we shall define $\xi(0) = -\infty$ and $\xi(1) = +\infty$.

For given integers m, $m \geqslant 2$ and i, $0 \leqslant i \leqslant m$, $\xi(i/m)$ will be called the i-th of m *fractile points* of F. $i = 0, 1, \ldots, m$ and the interval $\left(\xi\left(\frac{i-1}{m}\right), \xi\left(\frac{i}{m}\right)\right)$ will be called the i-th of m *fractile intervals* of F; $i = 1, 2, \ldots, m$. The conditional expectation of the random variable given that it belongs to the above interval will be denoted by $\mu(i, m)$ and called the i-th of m *fractile scores*.

Thus
$$\mu(i, m) = m \int_{\xi_{i-1}}^{\xi_i} x dF$$

where for simplicity
$$\xi_i = \xi\left(\frac{i}{m}\right).$$

Fractile scores for the standard normal distribution are available in Fisher and Yates (1938), Rao, Mitra, Matthai (1966) and other tables. Linder (1963) used these normal fractile scores in testing normality of a sample frequency distribution by methods of fractile graphical analysis developed by Mahalanobis (1958; 1960). Mahalanobis (1969) pointed out how fractile scores for other standard distributions can be used for similar purposes.

Evaluation of fractile scores associated with the gamma distribution reported in this paper was taken up at the suggestion of Professor Mahalanobis. We shall call these scores gamma scores: The one parmeter family of distribution functions considered is

$$G(x, \theta) = \int_{-\infty}^{x} g(t, \theta) dt$$

where

$$g(x,\theta) = \begin{cases} 0 & \text{for} \quad x < \theta \\ \frac{1}{\Gamma(\theta)} e^{-x}x^{\theta-1} & \text{for} \quad 0 \leqslant x < \infty \end{cases}$$

where $\theta > 0$ is the single parameter involved.

It is easy to see that the i-th of m fractile score for the gamma distribution is given by

$$\mu(i, m, \theta) = \theta m(G(\xi_i, \theta+1) - G(\xi_{i-1}, \theta+1))$$

where ξ_i is the i-th of m fractile points of $G(x, \theta)$. Since tables of $\mu(i, m, \theta)$, the gamma-scores, for a wide range of values of m and θ would be much too extensive for publication, we present here a computer program written in FORTRAN II language for calculation of these gamma-scores together with a description of the algorithms used.

2. Evaluation of $G(x, \theta)$

Let M be the largest integer not exceeding θ, $f = \theta - M$ and $u = 1+f$. We use the reduction formula

$$G(x,\theta) = \begin{cases} G(x, u) - g(x, u) \sum_{j=0}^{M-2} x^{j+1}/(u(u+1) \dots (u+j)), & \text{if } \theta > 2 \\ G(x, u), & \text{if } 1 < \theta \leqslant 2 \\ G(x, u) + g(x, u), & \text{if } 0 < \theta \leqslant 1 \end{cases}$$

We are thus led to consider evaluation of

$$A(x, u) = \int_0^x e^{-t} t^{u-1}\, dt$$

for $1 < u \leqslant 2$. For $x \leqslant 10$, expansion of e^{-t} in a power series and term by term integration is satisfactory, and we get for $A(x, u)$ the approximation

$$E(x, u) = \sum_{j=0}^{n} t(j, u, x)$$

where

$$t(j, u, x) = (-1)^j\, x^{u+j}/(j!(u+j))$$

and n is the smallest integer exceeding $2x$ for which $|t(n+1, u, x)| < 10^{-10}$. For $x > 10$, we derive an approximation $F(x, u)$ for

$$B(x, u) = \int_x^\infty e^{-t}\, t^{u-1} dt$$

by writing it in the form

$$B(x, u) = e^{-x} \int_0^\infty e^{-t}\, (t+x)^{u-1}\, dt$$

and using the third order Laguerre-Gauss quadrature formula; thus

$$F(x, u) = e^{-x} \sum_{k=1}^{3} H_k(x+x_k)^{u-1}$$

where H_k and x_k are respectively the weights and the abscissa of the quadrature formula borrowed here from Kopal (1961).

$$H_1 = 0.0103892565, \quad H_2 = 0.2785177336, \quad H_3 = 0.7110930099$$

$$x_1 = 6.289945083, \quad x_2 = 2.294280360, \quad x_3 = 0.4157745568$$

We then have the following approximation for $G(x, u)$:

$$J(x, u) = \begin{cases} E(x, u)/H(u) & \text{for } x \leqslant 10 \\ 1-F(x, u)/H(u) & \text{for } x > 10 \end{cases}$$

where
$$H(u) = E(10, u)+F(10, u)$$

is an approximation for $\Gamma(u)$. The algorithm presented above is an elaboration of one developed by Roy and Kalyanasundaram (1964) in a mimeographed report.

To evaluate fractile points of the gamma distribution we use an interative procedure based on the method of false position.

3. Error analysis

We obtain below estimate of errors in the approximations $E(x, u)$ for $A(x, u)$ and $F(x, u)$ for $B(x, u)$ without consideration of errors in rouding off.

Let

$$r_n(t) = (e^{-t} - \sum_{j=0}^{n} (-1)^j t^j/j!)\, t^{u-1}$$

$$= \sum_{j=n+1}^{\infty} (-1)^j t^{j+u-1}/j!$$

so that
$$|r_n(t)| \leqslant \frac{t^{n+u}}{(n+1)!} \left(1+\frac{t}{n+2}+\frac{t^2}{(n+2)(n+3)}+\ldots\right)$$

$$< \frac{t^{n+u}}{(n+1)!} \left(1+\frac{t}{n}+\left(\frac{t}{n}\right)^2+\ldots\right)$$

$$= \frac{t^{n+u}}{(n+1)!} \left(1-\frac{t}{n}\right)^{-1}$$

for $0 \leqslant t \leqslant x < n$. Thus, in this range

$$|r_n(t)| < \left(1-\frac{x}{n}\right)^{-1} \cdot \frac{x^{n+u}}{(n+1)!}$$

Therefore $|A(x, u)-E(x, u)| \leqslant \int_0^x |r_n(t)|\,dt$

$$< \left(1-\frac{x}{n}\right)^{-1} \frac{x^{n+1+u}}{(n+1)!\,(n+1+u)}$$

$$= \left(1-\frac{x}{n}\right)^{-1} |t(n+1, u, x)|$$

Now, if $n > 2x$ and $|t(n+1, u, x)| < 10^{-10}$ if follows that

$$|A(x, u)-E(x, u)| < 2\times 10^{-10}$$

Again, from the expression for the error term in Gauss-Laguerre quadrature formula (see Kopal, p. 373), we have

$$B(x, u)-F(x, u) = \frac{3!}{6!}\cdot\frac{d^6}{dt^6}\,(e^{-x}(t+x)^{u-1})\big|_{t=\eta}$$

where η is some positive number. It is thus easy to see that for $1 \leqslant u < 2$

$$|B(x, u)-F(x, u)| < e^{-x}x^{u-7}$$

so that if $x \geqslant 10$

$$|B(x, u)-F(x, u)| < e^{-10}\,10^{-5} < 5\times 10^{-10}$$

4. Fortran II program for gamma score

The list of a computer program in FORTRAN II language for evaluation of gamma scores, given the value of the parameter θ and the number of fractile groups m, is appended below. The program was tested on a $2K-4$ tape $H400$ computer system. It is a pleasure to acknowledge the assistance received from Mr. K. Vijayachandran in coding and debugging the program.

References

Fisher, R. A. and Yates, F. (1938): *Statistical Tables for Biological Agricultural and Medical Research*. Oliver and Boyd, London, 6th edition, 1963, 94.

Kopal, Z. (1961): *Numerical Analysis*; Chapman and Hall Ltd., London, 2nd edition; 373, 564-567.

Linder, A. (1963): Address delivered at the Second Convocation April 1963, Indian Statistical Institute.

Mahalanobis, P. C. (1958): A method for fractile graphical analysis with some surmises of results. *Transaction of the Bose Research Institute, Calcutta* 22, 223-230.

——— (1960): A method of fractile graphical analysis. *Econometrica*, 28, 325-354.

——— (1969): Extensions of fractile graphical analysis. *Paper presented at the International Conference on Quality Control, Tokyo, pre-print Indian Statistical Institute.*

Rao, C. R., Mitra, S. K. and Matthai, A. M. (1966): *Formulae and Tables for Statistical Work, Statistical Publishing Society, Calcutta*, 35.

Roy, J. and Kalyanasundaram, G. (1964): A Fortran Programme for Computing the Incomplete Gamma Function Ratio and its Inverse (Mimeographed Report, Indian Statistical Institute).

EVALUATION OF GAMMA SCORE

```
C             EVALUATION OF GAMMA SCORES
C             THAT IS MEANS OF N FRACTILE GROUPS OF G(X, V)
C             WHERE N, V ARE TO BE SPECIFIED BY USER
C
999           READ 1, N, V
  1           FORMAT (I2, F10.6)
              PRINT 250, N, V
250           FORMAT(10X, I2, 5X, F10.6)
              EN = N
              J = 1
              R = 1./EN
              PQ = 0.
              P = R
C
C
C             EVALUATION OF FRACTILES OF GAMMA DISTRIBUTION
C             THAT IS TO OBTAIN X SATISFYING P = G(X, V)
C
              E = 0.000001
              H = 0.1
              X1 = 0.
              P1 = 0.
        510   X2 = X1+H
              P2 = G(X2, V)
              IF (P2-P) 520, 570, 530
        520   X1 = X2
              P1 = P2
              GOTO510
        530   X = X1+(P-P1)*(X2-X1)/(P2-P1)
              PP = G(X, V)
              IF (ABSF(PP-P)-E) 580, 540, 540
        540   IF (PP-P)550, 580, 560
        550   X1 = X
              P1 = PP
              GOTO530
        560   X2 = X
              P2 = PP
              GOTO530
        570   X = X2
580           PR = G(X, V+1.)
        590   PM = V*EN*(PR-PQ)
              PRINT 600, J, X, PM
        600   FORMAT (5X, I2, 2F10.5)
              PQ = PR
              J = J+1
              P1 = P
              P = P+R
              X1 = X
              IF (J-N) 510, 610, 620
        610   PR = 1.0
              GOTO 590
620           PAUSE
              GO TO 999
              END
C
C
```

```
          FUNCTION G(X, V)
C
C
C         GAMMA FUNCTION
C         EVALUATION OF INCOMPLETE GAMMA INTEGRAL
C
          M = V
          EM = M
          U = V-EM+1.
          K = 1
          Y = 10.0
          GO TO 100
    5     H = S
          GO TO 300
   10     H = H+S
          K = 2
          Y = X
          IF (Y-10.) 100, 100, 300
   15     P = S/H
          GO TO 22
   20     P = 1.-S/H
   22     IF(V-2.) 25, 45, 200
   25     IF(V-1.) 30, 30, 45
   30     S = -1.
   32     Q = EXPF(-Y)*Y**(U-1.)/H
          G = P-Q*S
          RETURN
   45     G = P
          RETURN
  100     C = 0.0
          S = 0.
          T = Y**U/U
  105     S1 = S+T
          IF(S1-S) 110, 115, 110
  110     S = S1
          C = C+1.
          T = -T*(U+C-1.)*Y/(C*(U+C))
          GO TO 105
  115     GO TO (5, 15), K
  200     C = 0.0
          T = Y/U
          S = T
  205     IF (C-EM+2.) 210, 32, 32
  210     C=C+1.
          Y = T*Y/(U+C)
          S = S+T
          GO TO 205
  300     D = EXPF(-Y)
          S = D*(0.0103892565*((Y+6.269945083 )**(U-1.))
         1+0.2785177336*((Y+2.294280360)**(U-1.))
         2+0.7110930099*((Y+0.4157745568)**(U-1.)))
          GO TO (10, 20), K
          END
          END
          JOBEND
```

Paper received : *October*, *1970*.
Revised : *April*, *1971*.